




Maxiflex 1000 Series

Forth Programming for the



32 Channel Digital I/O Modules

Technical Note

TN1002

SCOPE: This Technical Note details the recommended Forth programming procedures for reading inputs from the M1326 32DI and writing outputs to the M1342 32DO Maxiflex Modules.

It is assumed that the reader is knowledgeable in the Ezi-Forth programming language as implemented on the existing range of Maxiflex CPU's.

(For further information on this subject see the Ezi-Forth Programmer's Manual)

DATE	REVISION	COMMENTS
23/04/98	01	Initial Publication. T1002.R01 32 Channel Digital Modules

INTRODUCTION

The most commonly used CPU's in the Maxiflex range are the M121x range generally known as the "Forth CPU's" because of the EziForth programming language resident on those CPU's.

This technical note describes to the use the recommended method of communicating in Forth with the M1326 and M1342 32 channel modules.

By way of example, the Forth code segments included later can actually be used to read and write digital I/O using these modules and the DIT.

I/O MODULE DATA REGISTERS

Each module has 3 data registers of meaningful information in them. These registers are read and written using the FROM and TO Forth commands. The data registers have the following layout:

Data Register Number	M1326 32DI	M1342 32DO
0	Scan Code and Module ID (1031)	Scan Code and Module ID (2058)
1	Digital Input status – channels 1 – 16	Digital Output status – channels 1 - 16
2	Digital Input status – channels 17 - 32	Digital Output status – channels 17 - 32

Register 0 is a unique identification number for each module i.e. 1031 for the M1326 32DI module. This number is always the same for the respective module. Registers 1 and 2 contain the current digital status of the module.

FORTH COMMANDS

- FROM:** read data from a register in the I/O module – requires the register number and the number of the I/O slot where the module is installed on the Maxiflex rack.
- TO:** write data to a register in the I/O module – requires the data, the register number and the number of the I/O slot where the module is installed on the Maxiflex rack.
- DITRD:** read from a register in the DIT of the CPU – requires the DIT register number to be read
- DITWR:** write to a register in the DIT of the CPU – requires the data and the DIT register number to be written.

The above Forth commands are not standard in the CPU. The example code segment shows how these commands are defined.

I/O REPEATER EXAMPLE

The following Forth application example reads the digital input status from the M1326 32DI module and writes it to DIT registers 200 and 201. Then the digital input status is read from the DIT registers 200 and 201 and written out to the M1342 32DO module, effectively repeating the input status. It is assumed that an M1326 32DI module is installed in slot 1 and an M1342 32DO is installed in slot 2. In addition, the example maps input channels 1 to 16 to bit 0 and bit 15 of DIT register 200 and input channels 17 to 32 to bit 0 and bit 15 of DIT register 201.i.e. channel 1 corresponds to bit 0, DIT 200 and channel 17 corresponds to bit 0, DIT 201.

Please refer to the Forth comments (written in brackets) for clarity of the code.

```
( Forth application example - Simple 32 Way digital input repeater )

( Declare some constants: )
200 CONSTANT INPUTSTATUS ( DIT registers 200 & 201 store the digital input status )
1031 CONSTANT 32DIMODULE ( Scan ID for M1326 32DI module )
2058 CONSTANT 32DOMODULE ( Scan ID for M1342 32DO module )

( Define FORTH commands: )
: FROM 32 BIOS ;
: TO 33 BIOS ;
: DITRD 5 BIOS ;
: DITWR 4 BIOS ;

( Read digital inputs at slot 1 if 32DI is installed )
: READDIGIN
0 1 FROM ( read scan code and module id from register 0, I/O slot 1 )
32DIMODULE = ( is the 32DI module in slot 1? )
IF
1 1 FROM ( read the first 16 inputs in register 1, I/O slot 1 )
DUP 256 / SWAP 256 * + ( map the inputs to bits in DIT )
INPUTSTATUS DITWR ( write the inputs to DIT 200 )
2 1 FROM ( read the remaining 16 inputs in register 2, I/O slot 1 )
DUP 256 / SWAP 256 * + ( map the inputs to bits in DIT )
INPUTSTATUS 1+ DITWR ( write the inputs to DIT 201 )
ENDIF
;

( Write digital outputs at slot 2 if 32DO is installed )
: READDIGOUT
0 2 FROM ( read scan code and module id from register 0, I/O slot 2 )
32DOMODULE = ( is the 32DO module in slot 2? )
IF
INPUTSTATUS DITRD ( read the first 16 inputs from DIT 200 )
DUP 256 / SWAP 256 * + ( re-map the inputs to outputs )
1 2 TO ( write the first 16 outputs to register 1, I/O slot 2 )
INPUTSTATUS 1+ DITRD ( read the remaining 16 inputs from DIT 201 )
DUP 256 / SWAP 256 * + ( re-map the inputs to outputs )
2 2 TO ( write the outputs to register 2, I/O slot 2 )
ENDIF
;

( Main loop to run the repeater code )
: MAIN
BEGIN ( start of an infinite loop )
READDIGIN ( read in the inputs and save to DIT )
READDIGOUT ( retrieve outputs from the DIT and write )
AGAIN ( loop back to start of loop )
;
```